Statistics methods 
extras by adc, 2004/5

Collection
not all of them work for all subclasses of Collection, but they do 
not depend on sequence, so I put them in Collection. 
E.g. it may be useful to have methods like

(a: 1, b: 2).stdDev;

sumF(function)

Like sum, answer the sum of the results of function evaluated for each item in the receiver.
The function is passed two arguments, the item and an integer index.
The sum is a Float to avoid number wraparound for large collections of large integers.

a = Array.rand(150000, 10000, 20000); 
a.sum.postln;		// sum fails here.
a.sumF.postln;


mean

Answer the mean value of the receiver. (this is part of the standard sc3 distribution.)

List[1, 2, 3, 4].mean;


meanF

Answer the mean value of the receiver; calls sumF internally 
for better handling of large collections of large integers.

a = Array.rand(150000, 10000, 20000); 
a.mean.postln;	// mean fails here because of .sum.
a.meanF.postln;

(1 .. 10).meanF;


geoMean

Answer the geometric mean value of the receiver.

[1, 2, 4, 8, 16].mean;
[1, 2, 4, 8, 16].geoMean;

harmMean

Answer the harmonic mean value of the receiver.

[1, 2, 4, 8, 16].mean;
[1, 2, 4, 8, 16].geoMean;
[1, 2, 4, 8, 16].harmMean;

[1, 2].mean;
[1, 2].geoMean;
[1, 2].harmMean;


variance

Answer the variance of the receiver.

[1,2,3,4].variance;

stdDev

Answer the standard deviation of the receiver.

[1,2,3,4].stdDev;

skew

Answer the skew of the receiver.

[1,2,3,4].skew;		// even
[1,2,4,8].skew;		// positive skew means more density at low values.
[0, 4, 6, 7].skew;

kurtosis

kurtosis is tails size : kurtosis > 0 is leptokurtic, i.e. large tails; 
0 is normal distribution, < 1 is platykurtic = small tails.
I am not quite sure how to test that this formula is correct, 
so if you have ideas, let me know.

(1..4).kurtosis;
(1..10).kurtosis;

zTable

Answer the zTable form of the receiver, i.e. scale the table 
such that its mean is 0 and its standard deviation is 1.

a = [1,2,3,4];
[ a.mean, a.stdDev ].postln;
b = [1,2,3,4].zTable.postln;
[ b.mean, b.stdDev ].postln;

histo (steps, min, max)

Answer a histogram of the values of the receiver; by dividing 
the range min - max into a given number of steps. 

a = Array.rand(500, 0.0, 1.0);
b = a.histo(100); 		// by default, get min and max from coll;
b.plot;

c = a.histo(100, 0, 1);	// if you supply a range, 
(b == c).postln;		// the result is often slightly different.

c.plot;			

SequenceableCollection

corr (aColl)

calculate the correlation between the receiver and a second SeqColl of equal size.

			// identical: corr is 1.
(0 .. 10).corr( (0..10) );	

			// second array is linear transform : also (nearly) (+-) 1.
(0 .. 10).corr((0 .. 10) * -0.01 + 5);

			// fails with NaN if one array is all equal elements.
[1,2,3,4].corr([1,1,1,1]);

			// random data arrays have some correlation value near zero.
Array.rand(1000, 0.0, 0.1).corr(Array.rand(1000, 0.0, 0.1));

